
Conceptos básicos

ASP es una tecnología desarrollada por MS para crear páginas web de contenido dinámico apoyándose en scripts ejecutados en el servidor. Básicamente una página ASP es una mezcla entre una página HTML y un programa que da como resultado una página HTML que es enviada al cliente (navegador).

Estos scripts o programas pueden en ASP ser escritos en uno de estos dos lenguajes de programación VBScript o JavaScript, pero el más extendido es VBScript, por eso centraremos todo este manual en VBScript y todas las posibilidades que él tiene.


ASP es una tecnología que pertenece a la parte servidor, por esto no es necesario que el cliente o navegador la soporte ya que se ejecuta en el servidor, sí que deberemos buscar un servidor que nos soporte este tipo de tecnología para que nuestras páginas corran correctamente.

Hay que destacar que ASP es una tecnología propietaria de Microsoft, y que el uso de esta tecnología implica el uso de los productos de Microsoft: MS Internet Information System y MS Windows en el servidor. 

Nuestro primer ASP

La ventaja que tiene ASP sobre otros lenguajes de programación que se ejecutan en el servidor (como podrían ser los script CGI Perl), es que nos permite intercalar las sentencias ASP en las paginas HTML, es un concepto algo complicado de entender si no se ha visto nunca como funciona unas paginas PHP o ASP. 

Vamos a ver un ejemplo sencillo para comprenderlo mejor. En azul está el código HTML y en verde el código ASP. Seguiremos este criterio durante todo el manual.

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 

<body> 

Parte de HTML normal. 
<BR><BR> 

<% 
   Dim i 
   Response.Write("Parte de ASP<br>") 
    
   for i=0 to 9 
      Response.Write("Linea " & i & "<br>") 
   next 
%> 

</body> 
</html> 

Ejecutar ejemplo Ver código
El código ASP ejecutado tiene dos partes: la primera imprime "Parte de ASP" y la segunda es un bucle que se ejecuta 10 veces de 0 a 9, por cada vez que se ejecuta se escribe una línea, la variable i contiene el número de línea que se está escribiendo. 

No importa si no entiende muy bien el programa este ejemplo solo es para ilustrar como se intercala el código HTML y el código ASP.

Sintaxis

La sintaxis del VBScript es muy parecida a la de Visual Basic, ya que de hecho VBScript es un subconjunto de Visual Basic. 

Las características más destacables son: 

· No hay distinción entre mayúsculas y minúsculas.

· Las instrucciones terminan con un retorno de carro.

· No es necesario definir las variables antes de usarlas, pero por claridad lo haremos.

· Las cadenas de texto se delimitan entre comillas dobles. "
· Los comentarios empiezan con una comilla simple ' y terminan al final de línea. 

Variables

Una variable es un contenedor de información, en el que podemos meter números enteros, números decimales, carácteres... el contenido de las variables se puede leer y se puede cambiar durante la ejecución de una página ASP. 

En ASP no es necesario definir las variables antes de usarlas. Tampoco tienen tipos, es decir que una misma variable puede contener un número y luego puede contener carácteres. 

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
   Dim a,b,c 
    
   a = 1 
   b = 3.34 
   c = "Hola Mundo" 
   Response.Write(a & "<br>" & b & "<br>" & c) 
%> 
</body> 
</html> 

Ejecutar ejemplo Ver código
En este ejemplo hemos definido tres variables, a, b y c y con la instrucción Response.Write hemos impreso el valor que contenían, insertando un salto de línea entre ellas.

Existen 2 tipos de variables, las variables locales que solo pueden ser usadas dentro de funciones y las variables globales que tienen su ámbito de uso fuera de las funciones. 

Operadores 

Aritméticos

Los operadores de VBScript son muy parecidos a los de Visual Basic puesto que es un lenguaje que procede de este último. Si usted conoce este lenguaje le resultaran familiares y fáciles de reconocer.

Estos son los operadores que se pueden aplicar a las variables y constantes numéricas.

	Operador
	Nombre
	Ejemplo
	Descripción

	+
	Suma
	5 + 6
	Suma dos números

	-
	Resta
	7 - 9
	Resta dos números

	*
	Multiplicación
	6 * 3
	Multiplica dos números

	/
	División
	4 / 8
	Divide dos números

	%
	Módulo
	7 mod 2
	Devuelve el resto de dividir ambos números, en este ejemplo el resultado es 1

	^
	Exponente
	8 ^ 4
	Eleva 8 a 4.


<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
   Dim a,b 
   a = 8 
   b = 3 
   Response.Write( a + b & "<br>") 
   Response.Write( a - b & "<br>") 
   Response.Write( a * b & "<br>") 
   Response.Write( a / b & "<br>") 
   Response.Write( a ^ b & "<br>") 
%> 
</body> 
</html> 

Comparación

Los operadores de comparación son usados para comparar valores y así poder tomar decisiones. 

	Operador
	Nombre
	Ejemplo
	Devuelve cierto cuando:

	=
	Igual
	a = b
	a es igual b

	<>
	Distinto
	a <> b
	a es distinto b

	<
	Menor que
	a < b
	a es menor que b

	>
	Mayor que
	a > b
	a es mayor que b

	<=
	Menor o igual
	a <= b
	a es menor o igual que b

	>=
	Mayor o igual
	a >= b
	a es mayor o igual que b


<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
   Dim a,b 
   a = 8 
   b = 3 
   c = 3 
   Response.Write( (a = b) & "<br>") 
   Response.Write( (a <> b) & "<br>") 
   Response.Write( (a < b) & "<br>") 
   Response.Write( (a > b) & "<br>") 
   Response.Write( (a >= c) & "<br>") 
   Response.Write( (b <= c) & "<br>") 
%> 
</body> 
</html> 

Lógicos

Los operadores lógicos son usados para evaluar varias comparaciones, combinando los posibles valores de estas.

	Operador
	Nombre
	Ejemplo
	Devuelve cierto cuando:

	and
	Y
	(7>2) and (2<4)
	Devuelve verdadero cuando ambas condiciones son verdaderas.

	or
	O
	(7>2) or (2<4)
	Devuelve verdadero cuando al menos una de las dos es verdadera.

	xor
	XOR
	(7>2) xor (2<4)
	Devuelve verdadero cuando solo una de las dos es verdadera.

	not
	No
	not (7>2)
	Niega el valor de la expresión.


<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
   Dim a,b,c 
   a = 8 
   b = 3 
   c = 3 
   Response.Write( ((a = b) and (c > b)) & "<br>") 
   Response.Write( ((a = b) or (b = c)) & "<br>") 
   Response.Write( (not (b <= c)) & "<br>") 
%> 
</body> 
</html> 

Condicionales

Las sentencias condicionales nos permiten ejecutar o no unas ciertas instrucciones dependiendo del resultado de evaluar una condición. Las más frecuentes son la instrucción if y la instrucción select.

Sentencia if ... else
<% 
   if condición then 
      Sentencias a ejecutar cuando la 
      condición es cierta. 
   else 
      Sentecias a ejecutar cuando la 
      condición es falsa. 
   end if 
%>
La sentencia if ejecuta una serie de instrucciones u otras dependiendo de la condición que le pongamos. Probablemente sea la instrucción más importante en cualquier lenguaje de programación.

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
   Dim a,b 
   a = 8 
   b = 3 
   if a < b then 
      Response.Write("a es menor que b") 
   else 
      Response.Write("a no es menor que b") 
   end if 
%> 
</body> 
</html> 

Ejecutar ejemplo Ver código
En este ejemplo la condición no es verdadera por lo que se ejecuta la parte de código correspondiente al else.

Sentencia select ... case
<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
   Dim posicion 
   posicion = "arriba" 
    
   select case posicion 
      case "arriba"   ' Bloque 1 
         Response.Write("La variable contiene") 
         Response.Write(" el valor arriba") 
      case "abajo"   ' Bloque 2 
         Response.Write("La variable contiene") 
         Response.Write(" el valor abajo") 
      case else   ' Bloque 3 
         Response.Write("La variable contiene otro valor") 
         Response.Write(" distinto de arriba y abajo") 
   end select 
%> 
</body> 
</html> 

Ejecutar ejemplo Ver código
Con la sentencia select case podemos ejecutar unas u otras instrucciones dependiendo del valor de una variable, en el ejemplo anterior, dependiendo del valor de la variable posicion se ejecuta el bloque 1 cuando el valor es "arriba", el bloque 2 cuando el valor es "abajo" y el bloque 3 si no es ninguno de los valores anteriores.

Bucles

Los bucles nos permiten iterar conjuntos de instrucciones, es decir repetir la ejecución de un conjunto de instrucciones mientras se cumpla una condición.

Sentencia while
<% 
   while condición 
      intrucciones a ejecutar. 
   wend 
%>
Mientras la condición sea cierta se reiterará la ejecución de las instrucciones que están dentro del while.

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
Inicio<BR> 
<% 
   Dim i 
   i=0 
   while i<10 
      Response.Write("El valor de i es " & i & "<br>") 
      i=i+1 
   wend 
%> 
Final<BR> 
</body> 
</html> 

Ejecutar ejemplo Ver código
En el siguiente ejemplo, el valor de i al comienzo es 0, durante la ejecución del bucle, se va sumando 1 al valor de i de manera que cuando i vale 10 ya no se cumple la condición y se termina la ejecución del bucle.

Sentencia for
<% 
   for variable=inicial to final 
      intrucciones a ejecutar. 
   next 
%>
<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
Inicio<BR> 
<% 
   Dim i 
   for i=0 to 9 
      Response.Write("El valor de i es " & i & "<br>") 
   next 
%> 
Final<BR> 
</body> 
</html> 

Ejecutar ejemplo Ver código
Con la instrucción for indicamos un rango de valores entre los que se moverá la variable de iteración, desde el valor indicado al principio hasta el valor que se indica al final, en este caso del 0 al 9. Dentro de la instrucción for se ejecutarán las instrucciones tomando i los valores del rango.

Salida

La única manera que tenemos en ASP para producir una salida es usando el objeto del servidor Response, este objeto tiene varios métodos y concretamente el método Write(cadena de texto) nos permite producir una salida que ira al navegador.

Response.Write
<% 
   Response.Write(cadena de texto) 
%>
Si la variable a imprimir no es de texto, tampoco hay ningún problema porque esta es convertida al instante a texto.

Existe también otra manera más corta de imprimir el valor de una variable, usando la forma reducida <%=variable%> que equivaldría a la instrucción <% Response.Write(variable) %>
<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
   Dim tex,num,fecha 
   tex="variable de texto" 
   num=45345 
   fecha=date 
   Response.Write("Texto simple<br>") 
   Response.Write(tex & "<br>") 
   Response.Write("Un numero:" & num & "<br>") 
   Response.Write("Una fecha:" & fecha & "<br>") 
%> 
   <%=num%> 
</body> 
</html> 

Manejo de cadenas

Dado el uso del lenguaje ASP el tratamiento de cadenas es muy importante, existen bastantes funciones para el manejo de cadenas, a continuación explicaremos las más usadas.

0. len(cadena). Nos devuelve el número de carácteres de una cadena. 

1. split(cadena,separador). Divide una cadena en varias usando un carácter separador. 

2. mid(cadena, inicio, longitud). Devuelve una subcadena de otra, empezando por inicio y de longitud longitud. 

3. trim(cadena). Elimina los espacios iniciales y finales de una cadena. 

4. instr(cadena1, cadena2). Busca la cadena2 dentro de cadena1 indicándonos la posición en la que se encuentra. 

5. replace(texto, cadena1, cadena2). Reemplaza la cadena1 por la cadena2 en el texto. 

6. left(cadena, num). Devuelve los num caracteres de cadena empezando por la izquierda. 

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
   Dim palabras,i 
   Response.Write(len("12345") & "<br>") 
   i=0 
   palabras=split("Esto es una prueba"," ") 

   Response.Write(palabras(0)&"<br>") 
   Response.Write(palabras(1)&"<br>") 
   Response.Write(palabras(2)&"<br>") 
   Response.Write(palabras(3)&"<br><br>") 

       

   Response.Write(mid("Devuelve una subcadena de otra",9,3) & "<br><br>") 

   if trim(" Cadena ") = "Cadena" then 
      Response.Write("Iguales<br><br>") 
   end if 

   Response.Write(instr("Busca la palabra dentro de la frase", "palabra") & "<br><br>") 
    
   Response.Write(replace("Un pez de color verde, como verde es la hierba.","verde","rojo") & "<br>") 
    
%> 
</body> 
</html> 

Procedimientos y Funciones

El uso de procedimientos y funciones nos da la capacidad de agrupar varias instrucciones bajo un solo nombre y poder llamarlas a estas varias veces desde diferentes sitios, ahorrándonos la necesidad de escribirlas de nuevo. 

Procedimiento

<% 
   sub Nombre(parametro1, parametro2,...) 
      instrucción1; 
      instrucción2; 
      instrucción3; 
      instrucción4; 
   end sub 
%>

Para llamar a un procedimiento tenemos dos sintaxis distintas:

Sin paréntesis 

Nombre parametro1, parametro2...

Con paréntesis 

call Nombre(parametro1, parametro2)
Las funciones son iguales que los procedimientos pero estas nos permiten devolver un valor.

Función

<% 
   function Nombre(parametro1, parametro2,...) 
      instrucción1; 
      instrucción2; 
      instrucción3; 
      instrucción4; 

      Nombre = Valor de retorno 
   end function 
%>
<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
    
   function media_aritmetica(a, b) 
      Dim media 
      media=(a + b)/2 
      media_aritmetica = media 
   end function 
    
   Response.Write(media_aritmetica(4,6) & "<br>") 
   Response.Write(media_aritmetica(3242,524543) & "<br>") 
    
%> 
</body> 
</html> 

Librerías

El uso de librerías es tremendamente útil, nos permiten agrupar varias funciones y variables en un mismo fichero, de manera que luego podemos incluir esta librería en distintas páginas y disponer de esas funciones fácilmente. 

<!-- Manual de ASP de WebEstilo.com --> 

<% 
   sub CabeceraPagina 
%> 
   <FONT SIZE="+1">Esta cabecera estará en todas sus páginas.</FONT><BR> 
   <hr> 
<% 
   end sub 
    
   sub PiePagina 
%>    
   <hr> 
   <FONT SIZE="-1">Este es el pie de página.</FONT><BR> 
   Autor: Joaquin Gracia 
<%    
   end sub 
%> 

Ver código
Ahora vamos a crear 2 páginas que usan la librería definida anteriormente para conseguir que las dos paginas tengan la misma cabecera y pie de pagina. 

La instrucción para incluir una librería en nuestra página es <!-- #include file="nombre de librería" -->
<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<!-- #include file="libreria01.asp" --> 
<% call CabeceraPagina %> 

Página 1 
<BR><BR><BR><BR><BR> 

Contenido blalbl blalb alb<BR><BR> 
más cosas...<BR><BR> 

fin<BR><BR> 

<% call PiePagina %> 
</body> 
</html> 

Ejecutar ejemplo Ver código
<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<!-- #include file="libreria01.asp" --> 
<% call CabeceraPagina %> 

Esta es otra página<BR><BR> 
completamente distinta<BR><BR> 
pero comparte el pie y la cabecera con la otra.<BR><BR> 

<% call PiePagina %> 
</body> 
</html> 

Envío y recepción de datos

El lenguaje ASP nos proporciona una manera sencilla de manejar formularios, permitiéndonos de esta manera procesar la información que el usuario ha introducido.

Al diseñar un formulario debemos indicar la página ASP que procesará el formulario, así como en método por el que se le pasará la información a la página.

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<H1>Ejemplo de procesado de formularios</H1> 
Introduzca su nombre: 
<FORM ACTION="procesa.asp" METHOD="GET"> 
<INPUT TYPE="text" NAME="nombre"><BR> 
<INPUT TYPE="submit" VALUE="Enviar"> 
</FORM> 
</body> 
</html> 

Ejecutar ejemplo Ver código
Al pulsar el botón Enviar el contenido de cuadro de texto es enviado a la página que indicamos en el atributo ACTION de la etiqueta FORM.

Para recoger los valores que han sido pasados a través de un formulario, tenemos que usar el objeto del servidor Request y dependiendo de como han sido pasado, si por GET o por POST, usaremos QueryString o Form. Si hemos usado el método GET, Request.QueryString("Nombre variable") y si hemos usado el método POST Request.Form("Nombre variable").

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<H1>Ejemplo de procesado de formularios</H1> 
El nombre que ha introducido es: <%=Request.QueryString("nombre")%> 
<br> 
</FORM> 
</body> 
</html> 

Método GET

En la página anterior hemos comentado que los datos de un formulario se envía mediante el método indicado en el atributo METHOD de la etiqueta FORM, los dos métodos posibles son GET y POST. 

Si usamos el método GET el datos son enviados mediante la URL y pueden ser vistos en esta. Para recogerlos deberemos usar Request.QueryString("Nombre variable")
<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<H1>Ejemplo de procesado de formularios</H1> 

<FORM ACTION="procesa2.asp" METHOD="GET"> 
Introduzca su nombre:<INPUT TYPE="text" NAME="nombre"><BR> 
Introduzca sus apellidos:<INPUT TYPE="text" NAME="apellidos"><BR> 
<INPUT TYPE="submit" VALUE="Enviar"> 
</FORM> 
</body> 
</html> 

Ejecutar ejemplo Ver código
procesa2.asp

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<H1>Ejemplo de procesado de formularios</H1> 
El nombre que ha introducido es: <%=Request.QueryString("nombre") & " " & Request.QueryString("apellidos") %> 
<br> 

</body> 
</html

Método POST

Ahora usaremos el método POST, a diferencia del método GET con el método POST los datos enviados no se ven en la barra del navegador. Para recogerlos deberemos usar Request.Form("Nombre variable")
<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<H1>Ejemplo de procesado de formularios</H1> 
<FORM ACTION="procesa3.asp" METHOD="POST"> 
Introduzca su nombre:<INPUT TYPE="text" NAME="nombre"><BR> 
Introduzca sus apellidos:<INPUT TYPE="text" NAME="apellidos"><BR> 
<INPUT TYPE="submit" VALUE="Enviar"> 
</FORM> 
</body> 
</html> 

Ejecutar ejemplo Ver código
procesa3.asp

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<H1>Ejemplo de procesado de formularios</H1> 
El nombre que ha introducido es: <%=Request.Form("nombre") & " " & Request.Form("apellidos") %> 
<br> 

</body> 
</html> 

Crear la base de datos

Para la realización de este curso sobre ASP con acceso a base de datos hemos elegido la base de datos Access por ser una base de datos muy conocida por todo el mundo, sin embargo si deseamos realizar alguna aplicación profesional recomendamos le uso de MS SQL Server, ya que Access se puede quedar pequeña rápidamente.

Una vez conseguido esto debemos crear las tablas en la base de datos, la descripción de las tablas contienen la estructura de la información que almacenaremos en ellas. Para lo cual usaremos en lenguaje de consultas SQL común para todas las bases de datos relacionales. 

Para estos sencillos ejemplos crearemos una base de datos que hemos llamado datos.mdb, y dentro de esta base de datos una tabla que hemos llamado prueba que tiene la siguiente estructura.

prueba.sql

CREATE TABLE prueba 
(id_Prueba COUNTER, 
Nombre varchar(100), 
Apellidos varchar(100))

En este ejemplo creamos una tabla llamada prueba con 3 campos: un campo identificador, que nos servirá para identificar unívocamente una fila con el valor de dicho campo, otro campo con el nombre de una persona y por último un campo con el apellido de la persona.

Conexión a la base de datos

Una vez que tenemos creada la base de datos en nuestro servidor, el siguiente paso es conectarnos a la misma desde una página ASP.

Para acceder a la base de datos usaremos ADO (ActiveX Data Objects), ADO son un conjunto de objetos que nos permiten acceder a la base de datos independientemente del motor de base de datos que usemos, así pues estos ejemplos usan MS Access pero funcionarían igual si el motor de Base de Datos fuese MS SQL Server. Tan solo habría que cambiar el driver. 

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<% 
Dim oConn 
    
Set oConn = Server.CreateObject("ADODB.Connection") 

oConn.Open("DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=" & Server.MapPath("\webestilo\db\datos.mdb")) 
    
   Response.Write("Conexión con la base de datos conseguida.<br>") 
    
oConn.Close 
set oConn = nothing 

%> 
</body> 
</html> 

Ejecutar ejemplo Ver código
Para conectarnos a la Base de datos creamos un objeto de ADO de tipo conexión para ello usamos Server.CreateObject, una vez que tenemos el objeto, le indicamos el driver que tiene que usar, en este caso el de Access y en donde se encuentra la base de datos.

Para cerrar la conexión con la base de datos usaremos el método Close y seguidamente como ya no usaremos el objeto lo destruimos asignándole nothing.

Consultas a la base de datos

Una vez que nos hemos conectado con el servidor de bases de datos, ya podemos realizar consultas a las tablas de la base de datos.

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<H1>Ejemplo de uso de bases de datos con ASP y ADO</H1> 
<% 
   Dim oConn,strSQL, objRS 
    
   Set oConn = Server.CreateObject("ADODB.Connection") 

   oConn.Open("DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=" & Server.MapPath("\webestilo\db\datos.mdb")) 
    
   strSQL = "SELECT Nombre, Apellidos FROM prueba" 
   Set objRS = oConn.Execute(strSQL) 
    
%> 
   <TABLE BORDER=1 CELLSPACING=1 CELLPADDING=1> 
      <TR><TD>&nbsp;Nombre</TD><TD>&nbsp;Apellidos&nbsp;</TD></TR> 
<%       

   while (not objRS.Eof) 
      Response.Write("<tr><td>&nbsp;" & objRS("Nombre") & "</td><td>&nbsp;" & objRS("Apellidos") & "&nbsp;</td></tr>") 
      objRS.MoveNext 
   wend 
    
   oConn.Close 
   set objRS = nothing 
   set oConn = nothing 

%> 
</table> 
</body> 
</html> 

Ejecutar ejemplo Ver código
En este ejemplo hemos ejecutado una consulta SQL a la base de datos con el método Execute del objeto conexión, esto nos devuelve un objeto de tipo RecordSet del cual podemos obtener los datos de la tabla.

El método EOF nos permite saber si hemos llegado al final del RecordSet, y el método MoveNext nos permite avanzar hacía adelante en el RecordSet, de esta manera recorremos todo el RecordSet mostrando los datos que este contiene. 

Y finalmente cerramos la conexión con la base de datos y destruimos el RecordSet y el objeto Connection.

Inserción de registros

Hasta ahora nos hemos conectado a una base de datos y hemos hecho consultas a la misma, ahora presentaremos como introducir nuevo registros en la base de datos.

Para ello usaremos un formulario y en el ACTION del FORM <FORM ACTION="programaASP"> indicaremos que debe ser procesado una pagina ASP, esta página lo que hará será introducir los datos del formulario en la base de datos.

ejem07d.asp

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<H1>Ejemplo de uso de bases de datos con ASP y ADO</H1> 
<form action="procesar.asp" method="post"> 
<TABLE> 
<TR> 
   <TD>Nombre:</TD> 
   <TD><INPUT TYPE="text" NAME="nombre" SIZE="20" MAXLENGTH="30"></TD> 
</TR> 
<TR> 
   <TD>Apellidos:</TD> 
   <TD><INPUT TYPE="text" NAME="apellidos" SIZE="20" MAXLENGTH="30"></TD> 
</TR> 
</TABLE> 
<INPUT TYPE="submit" NAME="accion" VALUE="Grabar"> 
</FORM> 
<hr> 
<% 
   Dim oConn,strSQL, objRS 
    
   Set oConn = Server.CreateObject("ADODB.Connection") 

   oConn.Open("DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=" & Server.MapPath("\webestilo\db\datos.mdb")) 
    
   strSQL = "SELECT Nombre, Apellidos FROM prueba" 
   Set objRS = oConn.Execute(strSQL) 
    
%> 
   <TABLE BORDER=1 CELLSPACING=1 CELLPADDING=1> 
      <TR><TD>&nbsp;Nombre</TD><TD>&nbsp;Apellidos&nbsp;</TD></TR> 
<%       

   while (not objRS.Eof) 
      Response.Write("<tr><td>&nbsp;" & objRS("Nombre") & "</td><td>&nbsp;" & objRS("Apellidos") & "&nbsp;</td></tr>") 
      objRS.MoveNext 
   wend 
    
   oConn.Close 
   set objRS = nothing 
   set oConn = nothing 

%> 
</table> 
</body> 
</html> 

Ejecutar ejemplo Ver código
procesar.asp

<% 
   Dim oConn,strSQL 
    
   Set oConn = Server.CreateObject("ADODB.Connection") 

   oConn.Open("DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=" & Server.MapPath("\webestilo\db\datos.mdb")) 
    
   strSQL = "insert into prueba (nombre, apellidos) values ('" & Request.Form("nombre") & "','" & Request.Form("apellidos") & "')" 
   oConn.Execute(strSQL) 
   oConn.Close 
   set oConn = nothing 
    
   Response.Redirect("ejem07d.asp") 
%> 

La primera página ASP ejem07d.asp es un formulario que nos permite introducir nombre y apellido para añadirlo a la base de datos, seguido de una consulta que nos muestra el contenido de la tabla prueba. El formulario llama a la pagina procesar.asp que añadirá los datos a la tabla.

La segunda página procesar.asp se conecta a la base de datos y añade un nuevo registro con la instrucción insert del lenguaje de base de datos SQL. Una vez el registro se ha añadido se vuelve a cargar la página ejem07d.asp
Borrado de registros

Y finalmente, para cerrar el ciclo, nos queda el borrado de registros. El borrado de registros el uno de los procesos más sencillos.

Para indicar que elemento vamos a borrar hemos usado un enlace a la página borra.asp pasándole el ID_Prueba de cada registro, de esta manera la página borra.asp sabe que elemento de la tabla ha de borrar.

ejem07e.asp

<!-- Manual de ASP de WebEstilo.com --> 
<html> 
<head> 
   <title>Ejemplo de ASP</title> 
</head> 
<body> 
<H1>Ejemplo de uso de bases de datos con ASP y ADO</H1> 
<% 
   Dim oConn,strSQL, objRS 
    
   Set oConn = Server.CreateObject("ADODB.Connection") 

   oConn.Open("DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=" & Server.MapPath("\webestilo\db\datos.mdb")) 
    
   strSQL = "SELECT * FROM prueba" 
   Set objRS = oConn.Execute(strSQL) 
    
%> 
   <TABLE BORDER=1 CELLSPACING=1 CELLPADDING=1> 
      <TR><TD>&nbsp;Nombre</TD> <TD>&nbsp;Apellidos&nbsp;</TD> <TD>Borrar</TD></TR> 
<%       

   while (not objRS.Eof) 
      Response.Write("<tr><td>&nbsp;" & objRS("Nombre") & "</td><td>&nbsp;" & objRS("Apellidos") & "&nbsp;</td><td><a href=""borra.asp?id=" & objRS("id_prueba") &""">borrar</a></td></tr>") 
      objRS.MoveNext 
   wend 
    
   oConn.Close 
   set objRS = nothing 
   set oConn = nothing 

%> 
</table> 
</body> 
</html> 

Ejecutar ejemplo Ver código
borra.asp

<% 
   Dim oConn,strSQL 
    
   Set oConn = Server.CreateObject("ADODB.Connection") 

   oConn.Open("DRIVER={Microsoft Access Driver (*.mdb)}; DBQ=" & Server.MapPath("\webestilo\db\datos.mdb")) 
    
   strSQL = "delete from prueba where id_prueba = " & Request.QueryString("id") 
   oConn.Execute(strSQL) 
   oConn.Close 
   set oConn = nothing 
    
   Response.Redirect("ejem07e.asp") 
%> 

La página borra.asp se conecta a la base de datos y borra el registro indicado en la variable id que ha sido pasado desde la página ejem07e.asp. Una vez el registro se ha borrado se vuelve a cargar la página ejem07e.asp
